Approssimazione della soluzione di un sistema lineare

col metodo di Jacobi

Enrico Bertolazzi

Carica le librerie

```
> initialize ;
  with(LinearAlgebra) :
  with(plots):
```

initialize

Definisce la procedura Normi

```
> Normi := proc(v)
    local i ;
    return max(seq(abs(v[i]),i=1..Dimension(v))) ;
  end proc :
```

Definisce la procedura Jacobi

```
> Jacobi := proc(A, b, xs, n, epsi)
    local i, j, res, x, dm, bf, rlist ;
    rlist := [] ;
    dm    := Dimension(b) ;
    x     := xs ;
    for i from 1 to n do
      res   := evalf(b - A.x) ;
      bf    := Normi(res) ;
      rlist := [ op(rlist), [i,bf] ] ;
      if bf < epsi then break end if ;
      for j from 1 to dm do
        x[j] := x[j] + evalf(res[j]/A[j,j]) ;
      end do ;
    end do ;
    return rlist, x ;
  end proc :
```

Esempio d'uso

```
> # definisce la matrice
  A := <<2,0,-2,1>|<0,2,0,-1>|<-1,0,1,0>|<-3,-1,0,1>> ;
```

$$A := \begin{bmatrix} 2 & 0 & -1 & -3 \\ 0 & 2 & 0 & -1 \\ -2 & 0 & 1 & 0 \\ 1 & -1 & 0 & 1 \end{bmatrix}$$

```
> # definisce il termine noto per avere soluzione <1,2,3,4>
  b := A.<1,2,3,4> ;
```

$$b := \begin{bmatrix} -13 \\ 0 \\ 1 \\ 3 \end{bmatrix}$$

```
> # approssima la soluzione con Jacobi
  rlist, sol := Jacobi(A, b, <0,3,2,1>, 50,1e-6) :
```

```
> # stampa la soluzione
  sol ;
```

$$\begin{bmatrix} 1.000488280 \\ 2.000244140 \\ 2.997558594 \\ 4.000854492 \end{bmatrix}$$

```
> # disegna la norma del residuo
  plot(rlist,style=line,thickness=2,color=blue,
       labels=["iterate","residuo"]);
```


duo
iterate
0
2
4
6
8
10
12
14
10
20
30
40
50

>